

1989年南宁市青少年计算机 程序设计竞赛试题及参考解答

罗海鹏 冼国光

(广西计算中心) (南宁市电教站)

摘 要

南宁市青少年计算机程序设计竞赛于1989年5月在南宁举行。本文列举了七道典型试题及其参考解答, 这些试题有一定难度并限定二小时内在APPLE II机上完成。

1. 一本书的页码是从1记到N的。把所有这些页码加起来的时候, 其中有一个页码被漏加了, 结果所得的和是1989。求这个漏加的页码是多少?

解:

把页码1, 2, 3, ……全部加起来, 一直加到超过1989为止。加得的和超过1989的部分, 即是漏加的页码。

变量意义:

P: 页码数,

S: 页码的和。

程序如下:

```
10 P = 1 : S = 1
20 P = P + 1
30 IF S < 1989 THEN S = S + P : GOTO 20
40 P = S - 1989
50 PRINT "P = ", P
60 END
RUN
P = 27
```

2. 真分数 $26/65$ 有一个有趣的性质, 即它的分母的十位数字和分子的个位数字相同, 并且把这两个相同的数字一起取消后, 得到分数 $2/5$, 而正好有 $26/65 = 2/5$, 相当于进行了约分。请找出分母小于100的所有这样的真分数。

解:

用二重循环来做。外循环变量是分母，从12考查到99；内循环变量是分子，从11考查到分母减1。然后按照题目要求做假约分工作，如果假约分和真约分的值是一样的，则输出这个结果。

变量意义：

I：分母；

J：分子；

A1、A2：分母的十位数数字和个位数数字；

B1、B2：分子的十位数数字和个位数数字。

程序如下：

```

10 FOR I=12 TO 99
20 A1=INT(I/10):A2=I-A1*10
30 IF A2=0 THEN 80
40 FOR J=11 TO I-1
50 B1=INT(J/10):B2=J-B1*10
60 IF (A1=B2) AND ABS(J/I-B1/A2)<0.001 THEN PRINT
    J, "/", I, "=", B1, "/", A2
70 NEXT J
80 NEXT I
100 END
RUN
16/64=1/4
26/65=2/5
19/95=1/5
49/98=4/8
    
```

3. 数列的第一项是2,第二项是3,第三项是4,第四项、第五项是前三项乘积的十位数字和个位数字(如果乘积有百位数字的话,则是数字的第四、五、六项),第六项、第七项是二、三、四项乘积的十位数字和个位数字,……。求这个数列的第1989项至2000项。

解：

用二个指针记住相乘的数的位置和新产生的数的位置。要注意乘积中含有零时的处理。

变量意义：

A：数组，装这个数列的各项，

I：指针，指向相乘的第一项，

J：指针，指向新产生的一项。

程序如下：

```

10 DIM A(2002)
20 A(1)=2:A(2)=3:A(3)=4:J=3
30 FOR I=1 TO 2000
    
```

```

40 IF J >= 2000 THEN 100
50 T$ = STR$(A(I) * A(I+1) * A(I+2))
60 IF LEN(T$) = 3 THEN J = J + 1 : A(J) = VAL(LEFT$(T$, 1)) :
    T$ = RIGHT$(T$, 2)
70 IF LEN(T$) = 2 THEN J = J + 1 : A(J) = VAL(LEFT$(T$, 1)) :
    T$ = RIGHT$(T$, 1)
80 J = J + 1 : A(J) = VAL(T$)
90 NEXT I
100 FOR I = 1989 TO 2000 : PRINT A(I); " "; : NEXT I
110 END
RUN
2 1 2 0 1 2 0 3 0 6 2 1

```

4. 求 $(11^{101} + 13^{103} + 15^{105}) / 17$ 的余数。

解:

分开求一个幂除以17的余数, 然后把三个余数加在一起再除以17求余数。11的n次幂除以17的余数求出来后, 11的n+1次幂除以17的余数只需拿刚才的余数乘以11再除以17来求。

变量意义:

R1, 11的幂除以17的余数;

R2, 13的幂除以17的余数;

R3, 15的幂除以17的余数;

R, 总的余数。

程序如下:

```

10 R1 = 1
20 T = 13 * 13 : R2 = T - INT(T / 17) * 17
30 T = 15 ^ 4 : R3 = T - INT(T / 17) * 17
40 FOR I = 1 TO 101
50 R1 = R1 * 11 : R1 = R1 - INT(R1 / 17) * 17
60 R2 = R2 * 13 : R2 = R2 - INT(R2 / 17) * 17
70 R3 = R3 * 15 : R3 = R3 - INT(R3 / 17) * 17
80 NEXT I
90 R = R1 + R2 + R3 : R = R - INT(R / 17) * 17
100 PRINT R
110 END
RUN
12

```

5. 对于所有十位数字和个位数字不相同的二位数(一位数的前面补一个零也当成二位数), 我们订出如下计算规则: 用这二位数的两个数字组成的较大数减去较小的数, 得到新

的二位数，对新的二位数依照上面的规则继续算下去，……。最后会发现，我们陷入了一个死循环里，或者说是跌入了一个数的黑洞里。用具体的例子来说明，例如38， $83 - 38 = 45$ ， $54 - 45 = 09$ ， $90 - 09 = 81$ ， $81 - 18 = 63$ ， $63 - 36 = 27$ ， $72 - 27 = 45$ ，……。 (45, 09, 81, 63, 27) 构成二位数的黑洞。从其他任何的二个数字不同的二位数开始，都会进入同一个黑洞中。求黑洞中每一个数接受进入黑洞的二位数的个数。注：由54进入黑洞和由45进入黑洞，我们认为是同样的。

解：

把黑洞中的各数存入数组。循环语句从1执行到49，按照题目的规则进行计算。每计算一次的结果，都和数组中的各数比较。最后的结果乘以2，就是每个黑洞数被跌入的频率数。

变量意义：

H1、H2：数组，装黑洞各数；

X：进入黑洞前的二位数的变化过程；

Z：二位数的十位数字；

W：二位数的个位数字；

Y：颠倒十位和个位的二位数；

S：数组，统计跌入黑洞的数的分布规律。

程序如下：

```

10 FOR I=1 TO 5 : READ H1(I), H2(I) : NEXT I
20 DATA 45, 54, 9, 90, 81, 18, 63, 36, 27, 72
30 FOR I=1 TO 49
40 X=I
45 Z=INT(X/10) : W=X-Z*10
50 IF Z=W THEN 110
60 FOR J=1 TO 5
70 IF X=H1(J) OR X=H2(J) THEN S(J)=S(J)+1 : GOTO 110
80 NEXT J
90 Y=W*10+Z : X=ABS(X-Y)
100 GOTO 45
110 NEXT I
120 FOR I=1 TO 5 : PRINT H1(I), S(I)*2 : NEXT I
130 END
RUN
45          18
9           18
81          18
63          18
27          18
    
```

6.按次序写出2的乘幂: 2, 4, 8, 16, 32, 64, 128, 256, 512, 1024, 2048, 4096, ……，可以注意到这些数的末位呈周期性重复出现, 2, 4, 8, 6, 2, 4, 8, 6, ……，周期为4。由2的幂组成的序列, 它们的末三位也会有一个类似的周期。求出这个周期, 并找出组成这个周期的数的序列。

解:

用数组存2的幂的末三位数。每产生一个新的幂, 都和数组中的数进行比较, 以求找出循环周期。

变量意义:

D: 装2的幂的末三位数序列;

P: 2的幂的末三位数;

N: 2的幂次;

S: 2的幂的末三位数重复的周期。

程序如下:

```

10 DIM D ( 400 )
20 P = 2 : N = 1 : D ( 1 ) = P
30 P = P + P
40 IF P > 1000 THEN P = P - 1000
50 FOR I = 1 TO N
60 IF P = D ( I ) THEN M = I : GOTO 100
70 NEXT I
80 N = N + 1 : D ( N ) = P
90 GOTO 30
100 S = 0
110 FOR I = M TO N : PRINT D ( I ), " ", : S = S + 1 : NEXT I
120 PRINT, PRINT "S = ", S
130 END
RUN
8 16 32 64 128 256 512 24 48 96 192 384 768 536 72 144 288 576 152 304
608 216 432 864 728 456 912 824 648 296 592 184 368 736 472 944 888 776
552 104 208 416 832 664 328 656 312 624 248 496 992 984 968 936 872
744 488 976 952 904 808 616 232 464 928 856 712 424 848 696 392 784
568 136 272 544 88 176 352 704 408 816 632 264 528 56 112 224 448 896
792 584 168 336 672 344 688 376 752 504
S = 100

```

7.城市某区域的街道图如右, 问从甲地走到乙地可以有多少条最短路线?

解: 向上走用1表示, 向右走用2表示。总共有七条街, 用七重循环来做, 注意其中有两
条街是不能走的。

变量意义:

I₁、I₂、…、I₇：表示从甲到乙要走的7条街，I取值为1时表示向上走，取值为2时表示向右走。


N：计合格的路线数；

B：计向右走的街道数。

程序如下：

```

10 N=0
20 FOR I1=1 TO 2
30 IF I1=2 THEN B=1
40 FOR I2=1 TO 2
50 IF I2=2 THEN B=B+1
60 IF I2=1 AND I1=2 THEN 290
70 FOR I3=1 TO 2
80 IF I3=2 THEN B=B+1
90 FOR I4=1 TO 2
100 IF I4=2 THEN B=B+1
110 IF B>3 THEN 270
120 IF (I1=1 AND I4=2) AND ((I2=1 AND I3=2) OR (I2=2
    AND I3=1)) THEN 270
130 FOR I5=1 TO 2
140 IF I5=2 THEN B=B+1
150 IF B>3 THEN 260
160 FOR I6=1 TO 2
170 IF I6=2 THEN B=B+1
180 IF B>3 THEN 250
190 FOR I7=1 TO 2
200 IF I7=2 THEN B=B+1
210 IF B<>3 THEN 240
220 N=N+1
230 PRINT N, ":", I1, I2, I3, I4, I5, I6, I7
240 NEXT I7 : B=B-1
250 NEXT I6 : B=B-1
260 NEXT I5 : B=B-1
270 NEXT I4 : B=B-1
280 NEXT I3 : B=B-1
290 NEXT I2 : B=B-1
300 NEXT I1 : END
RUN
1 : 1111222
    
```



2 : 1112122	11 : 1221112
3 : 1112212	12 : 1221121
4 : 1112221	13 : 1221211
5 : 1121122	14 : 1222111
6 : 1121212	15 : 2211112
7 : 1121221	16 : 2211121
8 : 1211122	17 : 2211211
9 : 1211212	18 : 2212111
10 : 1211221	19 : 2221111

THE QUESTIONS AND REFERENCE SOLUTIONS OF NANNING YOUNGSTERS PROGRAMMING IN 1989

Luo Haipeng

(*Computer centre of Guangxi*)

Xian Guoguang

(*Audie Visual Educational Centre of Nanning*)

ABSTRACT

Nanning youngsters competition of computer knowledge including programming was held in Nanning in May, 1989. In this article, we present seven questions of the programming competition which mostly involve numbers of specific property and reference solutions. These questions are moderately difficult and prescribe two hours for fulfilment on the APPLEII computer.